

Лабораторная работа № 4. Таймеры микроконтроллеров ATx8515

Цель работы:

- изучение структур и режимов работы таймеров и их программирование,
- анализ схем включения таймеров для проведения исследований,
- программирование задач с таймером.

Введение

Микроконтроллеры AVR в зависимости от класса (Tiny, Classic, Mega) и типа модели имеют в своём составе от одного до трёх таймеров/счётчиков общего назначения T0, T1 и T2.

Первый таймер (8-разрядный T0), имеющийся во всех моделях, может использоваться для отсчёта и измерения временных интервалов или как счётчик внешних событий, а в модели ATmega8515 еще и для сравнения с заданным значением. При переполнении счётного регистра таймера генерируется запрос на прерывание. Два других таймера (16-разрядный T1 и 8-разрядный T2) имеют, кроме уже названных, дополнительные функции. Оба таймера могут генерировать запрос на прерывание не только при переполнении счётного регистра, но и при наступлении ряда других событий. Они могут также использоваться в качестве широтно-импульсных модуляторов. Кроме того, таймер T2 может работать в асинхронном (относительно тактового сигнала микроконтроллера) режиме.

Работа таймеров микроконтроллеров ATx8515 описана ниже. Каждый таймер/счётчик использует один или более выводов микроконтроллера. Эти выводы могут быть либо линиями портов ввода/вывода с альтернативными функциями, либо выделенными выводами микроконтроллера. Все выводы микроконтроллеров ATx8515, относящиеся к таймерам/счётчикам, и их функции приведены в табл.1.

Таблица 1. Выводы, используемые таймерами/счётчиками общего назначения

| Название | ATx8515 | STK500 | Описание |
|----------|-----------|--------|----------------------------------|
| T0 | PB0 | PB0 | Вход внешнего сигнала таймера T0 |
| T1 | PB1 | PB1 | Вход внешнего сигнала таймера T1 |
| ICP | ICP/PE0* | PE0 | Вход захвата таймера T1 |
| OC1A | PD5 | PD5 | Выход схемы сравнения таймера T1 |
| OC1B | OC1B/PE2* | PE2 | Выход схемы сравнения таймера T1 |

*) в микроконтроллере AT90S8515 – выделенные выводы ICP и OC1B, в ATmega8515 - выводы порта PE0 и PE2.

При использовании линий портов ввода/вывода необходимо сконфигурировать эти выводы в соответствии с их функциональным назначением (вход или выход).

У всех микроконтроллеров семейства AVR имеется также сторожевой таймер, который является неизменным атрибутом всех современных микроконтроллеров. Этот таймер используется для предотвращения закливания программы.

1. Таймер/счётчик T0 микроконтроллера AT90S8515

Таймер/счётчик T0 (8-разрядный) может использоваться для формирования временных интервалов или для подсчёта числа внешних событий. Структурная схема таймера/счётчика T0 микроконтроллера AT90S8515 приведена на рис.1.




Рис.1. Структурная схема таймера/счётчика T0

Таймер содержит базовый счётчик TCNT0, регистр управления TCCR0 и схему управления таймером. Кроме того, в его состав входят по одному разряду регистра запросов прерываний TIFR и маски прерываний TIMSK.

Счётчик TCNT0 доступен в любой момент времени как для чтения, так и для записи. При записи в счётчик TCNT0 во время его работы счёт будет продолжен в следующем за командой записи машинном цикле. После подачи напряжения питания счётчик TCNT0 принимает нулевое состояние.

При переходе таймера/счётчика TCNT0 из состояния «\$FF» в состояние «\$00» устанавливается в «1» флаг переполнения TOV0 в регистре TIFR и генерируется запрос на прерывание. Разрешение прерывания осуществляется установкой в «1» разряда TOIE0 (бит 1)

регистра маски TIMSK. Флаг общего разрешения прерывания I в регистре состояния SREG микроконтроллера также должен быть установлен в «1».

Таймер/счётчик T0 может работать в двух режимах:

- 1) таймера; в этом режиме на вход таймера/счётчика поступают импульсы непосредственно тактового сигнала микроконтроллера СК или с выхода предделителя схемы управления;
- 2) счётчика событий; в этом режиме инкремент содержимого счётчика производится по активному фронту сигнала на входе T0 микроконтроллера (линия порта PB0).

Выбор режима работы (источника тактового сигнала), а также запуск и останов таймера/счётчика осуществляются с помощью разрядов CS02...CS00 регистра управления таймером TCCR0 (табл.2). Соответствие между состоянием этих разрядов и режимом работы таймера/счётчика приведено в табл.3. Остальные разряды регистра доступны только для чтения и содержат «0».

Таблица 2. Формат регистра TCCR0

| № разряда | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|-----------|---|---|---|---|---|------|------|------|
| Имя | - | - | - | - | - | CS02 | CS01 | CS00 |

Таблица 3. Выбор источника тактового сигнала для таймера/счётчика T0

| CS02 | CS01 | CS00 | Источник тактового сигнала |
|------|------|------|--|
| 0 | 0 | 0 | Таймер/счётчик остановлен |
| 0 | 0 | 1 | СК (тактовый сигнал микроконтроллера) |
| 0 | 1 | 0 | СК/8 |
| 0 | 1 | 1 | СК/64 |
| 1 | 0 | 0 | СК/256 |
| 1 | 0 | 1 | СК/1024 |
| 1 | 1 | 0 | Вывод T0, инкремент счётчика производится по спадающему фронту импульсов |
| 1 | 1 | 1 | Вывод T0, инкремент счётчика производится по нарастающему фронту импульсов |

При использовании таймера/счётчика в режиме счёта внешних событий необходимо помнить, что сигнал, присутствующий на выводе T0, синхронизируется частотой тактового генератора микроконтроллера (состояние вывода T0 считывается по нарастающему фронту внутреннего тактового сигнала). Поэтому для обеспечения корректной работы таймера от

внешнего сигнала промежутки времени между соседними импульсами должен быть больше периода тактового сигнала микроконтроллера.

Инкремент содержимого таймера/счётчика при работе в режиме счёта внешних событий производится даже в том случае, если вывод T0 сконфигурирован как выход. Эта особенность даёт пользователю возможность программно управлять процессом счёта.

2. Таймер/счётчик T1 микроконтроллеров ATx8515

Таймер/счётчик T1 (16-разрядный) имеет гораздо больше функций, чем таймер/счётчик T0. Прежде всего, как и таймер/счётчик T0, он может использоваться для формирования временных интервалов или для подсчёта числа внешних событий на входе T1 (линия порта PB1). Во-вторых, таймер/счётчик T1 может при поступлении внешнего сигнала сохранять своё текущее состояние в отдельном регистре. В-третьих, он может выполнять определённые действия при равенстве содержимого счётного регистра и заданного значения. И, наконец, он может работать как широтно-импульсный модулятор (ШИМ). Следует иметь в виду, что генерация сигнала ШИМ «вынесена» в отдельный режим работы таймера/счётчика, в котором недоступны остальные функции (кроме генерации прерываний). В дальнейшем режим генерации сигнала ШИМ будем называть режимом ШИМ, а режим, в котором доступны остальные функции таймера/счётчика, - режимом таймера.

Структурная схема таймера/счётчика T1 приведена на рис. 2. В состав таймера/счётчика входят базовый 16-разрядный счётчик TCNT1, три 16-разрядных регистра (регистр захвата ICR1 и два регистра сравнения OCR1A и OCR1B), два 16-разрядных компаратора (схемы сравнения), два 8-разрядных управляющих регистра TCCR1A и TCCR1B, а также блок управления таймером.

Все флаги состояния таймера/счётчика (переполнения, совпадения и захвата) находятся в регистре флагов прерываний от таймеров TIFR, а разрешение/запрещение прерываний от таймера осуществляется установкой/сбросом соответствующих разрядов регистра маски TIMSK.

Базовый 16-разрядный счётчик TCNT1 реализован как суммирующий (в режиме ШИМ – как суммирующий/вычитающий) счётчик и доступен в любой момент времени как для чтения, так и для записи. При записи в счётчик TCNT1 во время работы таймера счёт будет продолжен по следующему за операцией записи импульсу тактового сигнала таймера/счётчика. После подачи напряжения питания счётчик TCNT1 принимает нулевое состояние.

Физически счётчик TCNT1 размещён в двух регистрах TCNT1H:TCNT1L. Чтобы при обращении процессорного устройства микроконтроллера к этим регистрам процесс записи (чтения) обоих байтов содержимого таймера/счётчика происходил “одновременно”, обращение производится с использованием специального 8-разрядного регистра TEMP. Этот регистр используется только процессором и программно недоступен.




Рис. 2. Структурная схема таймера/счётчика T1

Запись и чтение регистра TCNT1 происходят следующим образом:

- 1) при записи старшего байта значения в регистр TCNT1H он помещается в регистр TEMP. Далее, при записи младшего байта в регистр TCNT1L он объединяется с содержимым регистра TEMP, и оба байта записываются в регистр TCNT1 одновременно. Из сказанного следует, что для выполнения полного цикла записи в 16-разрядный счетчик первым должен загружаться старший байт (регистр TCNT1H).
- 2) при чтении регистра TCNT1L (младший байт) содержимое регистра TCNT1H пересылается в регистр TEMP. При последующем чтении регистра TCNT1H возвращается значение, сохранённое в регистре TEMP. Следовательно, для выполнения полной операции чтения 16-разрядного счетчика первым должен считываться младший байт (регистр TCNT1L).

Подобным образом с использованием регистра TEMP осуществляется обращение к остальным 16-разрядным регистрам: OCR1A, OCR1B и ICR1. Прерывания на время обращения к любому из этих регистров должны быть запрещены.

Управление таймером/счётчиком T1 осуществляется с помощью двух 8-разрядных регистров управления TCCR1A и TCCR1B. Формат регистров приведён в табл.4. Значение отдельных разрядов этих регистров будет описано далее. Неиспользуемые разряды регистров

доступны только для чтения и содержат «0».

Таблица 4. Формат управляющих регистров TCCR1A, TCCR1B

| Регистр\бит | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|---------------|--------|--------|--------|--------|------|------|-------|-------|
| TCCR1A | COM1A1 | COM1A0 | COM1B1 | COM1B0 | - | - | PWM11 | PWM10 |
| TCCR1B | ICNC1 | ICES1 | - | - | CTC1 | CS12 | CS11 | CS10 |

По отношению к тактовому сигналу таймер/счётчик T1 может работать в двух режимах, аналогично таймеру T0. Выбор источника тактового сигнала, а также запуск и останов таймера/счётчика осуществляются с помощью разрядов CS12...CS10 регистра управления таймером TCCR1B. Соответствие между состоянием этих разрядов и режимом работы таймера/счётчика приведено в табл. 5.

Таблица 5. Выбор источника тактового сигнала для таймера/счётчика T1

| CS12 | CS11 | CS10 | Источник тактового сигнала |
|------|------|------|--|
| 0 | 0 | 0 | Таймер/счётчик остановлен |
| 0 | 0 | 1 | СК (тактовый сигнал микроконтроллера) |
| 0 | 1 | 0 | СК/8 |
| 0 | 1 | 1 | СК/64 |
| 1 | 0 | 0 | СК/256 |
| 1 | 0 | 1 | СК/1024 |
| 1 | 1 | 0 | Вывод T1, инкремент счётчика производится по спадающему фронту импульсов |
| 1 | 1 | 1 | Вывод T1, инкремент счётчика производится по нарастающему фронту импульсов |

Режим таймера

Принцип работы таймера/счётчика T1 в режиме таймера такой же, как и таймера/счётчика T0. По каждому импульсу, поступающему на тактовый вход таймера/счётчика, производится инкремент содержимого счётчика TCNT1. При переходе таймера/счётчика из состояния «\$FFFF» в состояние «\$0000» устанавливается флаг переполнения TOV1 регистра TIFR и генерируется запрос на прерывание. Разрешение прерывания осуществляется установкой в «1» разряда TOIE1 (бит 7) регистра маски TIMSK (разумеется, флаг общего разрешения прерываний I регистра SREG также должен быть установлен в «1»). Кроме основных функций – подсчёт числа импульсов тактового сигнала микроконтроллера и подсчёт числа внешних событий, в режиме таймера T1 доступны и дополнительные функции.

Функция захвата (capture)

Данная функция заключается в сохранении в определённый момент времени состояния таймера/счётчика TCNT1 в регистре захвата ICR1. Это действие может производиться либо по активному фронту сигнала на выводе ICP микроконтроллера (линия порта PE0), либо по сигналу от аналогового компаратора, которые определяют сигнал захвата на входе блока управления таймера (рис.2). При этом устанавливается флаг ICF1 регистра TIFR и генерируется запрос на прерывание. Разрешение прерывания осуществляется установкой в «1» разряда TICIE1 (бит 3) регистра TIMSK.

Для управления схемой захвата используются два разряда регистра TCCR1B: ICNC1 и ICES1. Разряд ICNC1 управляет схемой подавления помех. Если этот разряд сброшен в «0», схема подавления помех выключена и захват производится по первому же активному фронту на выводе ICP микроконтроллера. Если же этот разряд установлен в «1», то при появлении активного фронта на выводе ICP схема управления выполняет четыре выборки с частотой, равной тактовой частоте микроконтроллера. Захват будет выполнен только в том случае, если все выборки имеют уровень, соответствующий активному фронту сигнала захвата (уровень «1» - для нарастающего и уровень «0» - для спадающего).

Активный фронт сигнала, по которому будет выполнено сохранение содержимого счётчика TCNT1 в регистре захвата, определяется состоянием разряда ICES1. Если этот разряд сброшен в «0», то активным является спадающий фронт. Если же этот разряд установлен в «1», то активным является нарастающий фронт.

Физически регистр захвата ICR1 размещён в двух регистрах ICR1H:ICR1L, доступных только для чтения. Поскольку регистр захвата является 16-разрядным, при его чтении, как уже было сказано, используется специальный регистр TEMP. При чтении регистра ICR1L (младший байт) содержимое этого регистра посылается в процессорное устройство, а содержимое регистра ICR1H (старший байт) сохраняется в регистре TEMP. При чтении регистра ICR1H возвращается значение, сохранённое в регистре TEMP. Следовательно, при чтении регистра ICR1 первым должен быть прочитан регистр ICR1L. Прерывания на время обращения к регистру ICR1 должны быть запрещены.

Функция сравнения (compare)

Данная функция заключается в непрерывном (каждый машинный цикл) сравнении содержимого счётчика TCNT1 и регистра сравнения. При совпадении их содержимого устанавливается флаг соответствующего прерывания. В микроконтроллерах ATx8515 имеется два регистра сравнения (OCR1A и OCR1B), причём операция сравнения производится независимо для каждого регистра. Если значение счётчика становится равным значению в регистре сравнения, то в следующем машинном цикле устанавливается соответствующий этому регистру флаг

прерывания в регистре TIFR (для регистра OCR1A – флаг OCF1A, для регистра OCR1B – флаг OCF1B) и генерируется запрос на прерывание. Разрешение прерываний осуществляется установкой в «1» соответствующих разрядов регистра TIMSK (OCIE1A – бит 6 для запроса OCF1A и OCIE1B – бит 5 для запроса OCF1B).

Наряду с установкой флага в регистре TIFR при равенстве счётчика и регистра сравнения могут выполняться и другие действия: сброс таймера/счётчика (только при сравнении с OCR1A) и изменение состояния определённого вывода микроконтроллера (при обоих сравнениях).

Выполнение или невыполнение указанных действий определяется несколькими разрядами регистров управления TCCR1A и TCCR1B. Состояние разрядов COM1x1, COM1x0 (x=A,B) регистра TCCR1A определяет поведение вывода OC1x при совпадении содержимого счётчика TCNT1 и регистра сравнения OCR1x согласно табл. 6. При изменении состояния этих разрядов соответствующее прерывание от схемы компаратора рекомендуется запретить (во избежание ложной генерации прерывания). Чтобы таймер/счётчик мог управлять выводом, последний должен быть сконфигурирован как выход порта.

Таблица 6. Управление выводом OC1x (x=A,B)

| COM1x1 | COM1x0 | Описание |
|--------|--------|--|
| 0 | 0 | Таймер/счётчик T1 отключен от вывода OC1x |
| 0 | 1 | Состояние вывода меняется на противоположное |
| 1 | 0 | Вывод сбрасывается в «0» |
| 1 | 1 | Вывод устанавливается в «1» |

Если разряд CTC1 регистра управления TCCR1B установлен в «1», то при совпадении содержимого счётчика TCNT1 и регистра сравнения OCR1A производится сброс таймера/счётчика в нулевое состояние.

Каждый регистр сравнения физически размещается в двух регистрах ввода-вывода: OCR1A – OCR1AH:OCR1AL; OCR1B – OCR1BH:OCR1BL.

Поскольку регистры сравнения являются 16-разрядными, при их чтении и записи используется специальный регистр TEMP. Процесс записи и чтения 16-разрядных регистров выполняется так же, как и регистра TCNT1. Прерывания на время обращения к регистрам сравнения OCR1A и OCR1B должны быть запрещены.

Режим ШИМ (PWM)

Широтно-импульсная модуляция является одним из видов непрерывной импульсной модуляции, при которой ширина импульса пропорциональна значению модулирующего сигнала.

Соответственно в данном случае широтно-импульсная модуляция заключается в генерировании сигнала с программируемой частотой и скважностью.

Для перевода таймера/счётчика T1 в режим ШИМ и задания частоты ШИМ-сигнала используются разряды PWM11:PWM10 регистра управления таймером TCCR1A. Соответствие между состоянием этих разрядов и режимом работы таймера/счётчика T1 приведено в табл.7.

Таблица 7. Управление режимом ШИМ таймера/счётчика T1

| PWM11 | PWM10 | Описание |
|-------|-------|---|
| 0 | 0 | Режим ШИМ таймера/счётчика выключен |
| 0 | 1 | 8-разрядный широтно-импульсный модулятор |
| 1 | 0 | 9-разрядный широтно-импульсный модулятор |
| 1 | 1 | 10-разрядный широтно-импульсный модулятор |

Для генерации сигнала ШИМ используется схема сравнения таймера/счётчика, поэтому в микроконтроллерах ATx8515 модулятор является двоянным (два регистра сравнения). Названия регистров сравнения и правила обращения к ним были описаны выше. Сигнал снимается с выхода схемы сравнения таймера/счётчика.

В рассматриваемом режиме счётчик TCNT1 функционирует как реверсивный счётчик, модуль счёта которого (TOP) зависит от режима работы модулятора. Частота ШИМ-сигнала зависит от частоты тактового сигнала f_{TCK1} таймера/счётчика T1 и модуля счёта ШИМ. Значение модуля счёта и частота ШИМ-сигнала для каждого режима работы модулятора приведены в табл.8.

Таблица 8. Режимы ШИМ

| Режим модулятора | Модуль счёта (TOP) | Частота ШИМ-сигнала |
|------------------|--------------------|---------------------|
| 8-разрядный | 255 | $f_{TCK1}/510$ |
| 9-разрядный | 511 | $f_{TCK1}/1022$ |
| 10-разрядный | 1023 | $f_{TCK1}/2046$ |

При работе таймера/счётчика T1 в режиме ШИМ состояние счётчика изменяется от 0 до значения TOP, а затем снова до 0, после чего цикл повторяется. При равенстве содержимого счётчика и содержимого регистра сравнения состояние соответствующего этому регистру вывода микроконтроллера изменяется согласно табл.9 (x обозначает А или В). Таким образом, длительность неинвертированного ШИМ-сигнала равна $2n/f_{TCK1}$, где n - содержимое регистра

сравнения.

Таблица 9. Поведение выходов схемы сравнения в режиме ШИМ

| COM1x1 | COM1x0 | Поведение вывода OC1x |
|--------|--------|---|
| 0 | 0 | Таймер/счётчик T1 отключен от вывода |
| 0 | 1 | Таймер/счётчик T1 отключен от вывода |
| 1 | 0 | Сбрасывается в «0» при прямом счёте и устанавливается в «1» при обратном счёте (неинвертированный ШИМ-сигнал) |
| 1 | 1 | Устанавливается в «1» при прямом счёте и сбрасывается в «0» при обратном счёте (инвертированный ШИМ-сигнал) |

Если в регистр сравнения записать значение «0» или TOP, то при следующем совпадении содержимого счётчика и содержимого регистра сравнения выход схемы сравнения переключится в устойчивое состояние согласно табл.10 (x = A или B).

Таблица 10. Устойчивые состояния выхода схемы сравнения

| COM1x1 | COM1x0 | Регистр OCR1x | Состояние вывода OC1x |
|--------|--------|---------------|-----------------------|
| 1 | 0 | 0 | 0 |
| 1 | 0 | TOP | 1 |
| 1 | 1 | 0 | 1 |
| 1 | 1 | TOP | 0 |

Особенностью работы таймера/счётчика T1 в режиме ШИМ является то, что при записи в регистр сравнения младшие 8 разрядов записываемого числа на самом деле сохраняются в специальном временном регистре. Изменение содержимого регистра сравнения происходит только в момент достижения счётчиком максимального значения TOP. Благодаря такому решению исключается появление в сигнале ШИМ импульса со случайной длительностью.

Соответственно, при чтении регистра сравнения в промежутке между записью в него и его действительным изменением возвращается содержимое временного регистра, то есть всегда возвращается значение, записанное последним.

При работе таймера/счётчика T1 в режиме ШИМ может генерироваться прерывание по переполнению счётчика, а также прерывание от схемы сравнения. Флаги прерываний устанавливаются в «1» при изменении счётчиком направления счёта: флаг TOV1 – в точке «0», а флаги OCF1A (для регистра OCR1A) и OCF1B (для регистра OCR1B) – в точке TOP. Разрешение и обработка соответствующих прерываний выполняются как обычно.

3. Программирование таймеров

Режим счетчика

Подготовим программу для исследования таймера/счётчика T0 в режиме счёта событий. Событием в данном случае будет нажатие одной из кнопок на плате STK500, подключенной к входу таймера T0. Результат работы программы отобразим средствами индикации.

Соединим кнопку SW0 с выводом порта PB0 (вход таймера T0). С помощью 10-проводного шлейфа соединим светодиоды LED0-LED7 с выводами порта PD. Запрограммируем вывод PB0 на ввод, все выводы порта PD – на вывод.

Настраиваем таймер на режим счёта внешних событий (от кнопки SW0). После отсчета нескольких, например четырёх, нажатий кнопки включим светодиоды на время 0,5 с и восстановим таймер для нового отсчета.

Для контроля числа нажатий воспользуемся схемой формирования запроса на прерывания, который вырабатывается при переполнении таймера в процессе счета. Для того, чтобы переполнение происходило после отсчета 4-х входных сигналов, необходимо перед началом работы загрузить в таймер начальное значение $TCNT0 = (2^8 - 4) = 252 = 0xFC$.

Вызываемый обработчик запроса прерывания включает светодиоды, выполняет подпрограмму задержки *delay* на время свечения, затем выключает светодиоды и обновляет состояние таймера для повторного отсчета событий.

Программа 4. 1

```
;*****
;Программа 4.1 для МК ATx8515:
;демонстрация работы таймера T0 в режиме счётчика событий;
;событие - нажатие кнопки SW0.
;Соединения: порт PB0-SW0, шлейфом порт PD-LED
;Светодиоды включаются после четвертого нажатия кнопки SW0
;*****
#include "8515def.inc"           ;файл определений AT90S8515
#include "m8515def.inc"         ;файл определений ATmega8515
.def temp = r16                 ;временный регистр
;***Таблица векторов прерываний
.org $000
        rjmp INIT               ;обработка сброса
.org $007
        rjmp T0_OVF             ;обработка переполнения таймера T0
```

```
***Инициализация МК
INIT:      ldi temp,low(RAMEND) ;установка
           out SPL,temp        ; указателя стека
           ldi temp,high(RAMEND) ; на последнюю
           out SPH,temp        ; ячейку ОЗУ
           clr temp            ;инициализация выводов порта PB
           out DDRB,temp       ; на ввод
           ldi temp,(1<<PB0)   ;включение 'подтягивающего' резистора
           out PORTB,temp      ; входа PB0
           ser temp            ;инициализация выводов порта PD
           out DDRD,temp       ; на вывод
           out PORTD,temp      ;выключение светодиодов
           ldi temp,(1<<SE)     ;разрешение перехода
           out MCUCR,temp      ; в режим Idle
           ;***Настройка таймера T0 на режим счётчика событий
           ldi temp,0x02       ;разрешение прерывания по
           out TIMSK,temp      ; переполнению таймера T0
           ldi temp,0x07       ;переключение таймера
           out TCCR0,temp      ; по положительному перепаду напряжения
           sei                 ;глобальное разрешение прерываний
           ldi temp,0xFC       ;$FC=-4 для
           out TCNT0,temp      ; отсчёта 4-х нажатий
LOOP:      sleep              ;переход в режим пониженного
           nop                 ; энергопотребления
           rjmp LOOP
           ;***Обработка прерывания при переполнении таймера T0
T0_OVF:    clr temp
           out PORTD,temp      ;включение светодиодов
           rcall DELAY         ;задержка
           ser temp
           out PORTD,temp      ;выключение светодиодов
           ldi temp,0xFC       ;перезагрузка
           out TCNT0,temp      ; TCNT0
           reti
           ;*** Задержка ***
DELAY:     ldi r19,6
           ldi r20,255
```

```
        ldi r21,255
dd:      dec r21
        brne dd
        dec r20
        brne dd
        dec r19
        brne dd
        ret
```

Результат работы программы будет таков. При четвёртом нажатии на кнопку SW0 загораются все светодиоды (в случае “дребезга” контактов светодиоды могут включиться раньше). Длительность времени, в течение которого они горят, определяется задержкой DELAY. Далее действия могут быть повторены.

Как уже было сказано в теоретической части, события для таймера/счётчика T0 можно генерировать программно. Для этого необходимо настроить вывод PB0 как выход. Увеличение содержимого счётчика будет происходить после выполнения команд программы, эмулирующих положительный или отрицательный перепад, как этого требует настройка таймера:

```
;положительный перепад напряжения (0/1) на PB0
        cbi PORTB,0
        sbi PORTB,0
;отрицательный перепад напряжения (1/0) на PB0
        sbi PORTB,0
        cbi PORTB,0
```

Задание 1. Проверить на плате STK500 работу исходной программы 4.1.

Изменить программу, исключив влияние на работу таймера возможность “дребезга” кнопки. Для этого запрограммировать линию порта PB0 на вывод для программного ввода в таймер положительного сигнала при замыкании кнопки. Формирование сигнала «0/1» для таймера выполнить, предварительно проверив состояние кнопки командой `sbi PINx,b` (порт и линию для кнопки выбрать самостоятельно, настроив ее на ввод и подключив подтягивающий резистор). При замыкании кнопки (сигнале «0») программно сформировать сигнал «0/1» на выходе PB0, в противном случае процессор остается в цикле ожидания.

Так как новый ввод в таймер можно производить только после отпускания кнопки (сигнала «1»), в программу следует добавить команды задержки `rcall DELAY` и проверки отпускания кнопки `sbi PINx,b`. (Примечание. При программном управлении работой таймера режим **sleep**

должен быть отменен).

Перед загрузкой программы выполнить ее отладку в AVR Studio. Убедившись в правильности работы программы, проверить ее на плате STK500.

Режим таймера

Подготовим программу для исследования 16-разрядного таймера/счётчика T1 в режиме таймера (отсчет тактовых импульсов). Поскольку счетчик работает в режиме суммирования, то после начальной загрузки значения TCNT1 счетчик выполняет счет до максимального значения 65535 (0xFFFF), после чего обнуляется (0x0000). Формируемый при этом сигнал переполнения счетчика, можно использовать в качестве запроса на прерывание основной программы, используя его, например, для перезагрузки таймера значением TCNT1. Отсчитываемый до переполнения интервал $N = (65536 - TCNT1)$ выполняется за время T :

$$T = N \cdot T_{cnt} = N / F_{cnt} = N \cdot K / F_{ск} = (65536 - TCNT1) \cdot K / F_{ск},$$

где T_{cnt} и F_{cnt} – период и частота следования счетных импульсов,

K – коэффициент делителя рабочей частоты микроконтроллера $F_{ск}$,

$TCNT1$ – начальное значение, загружаемое в 16-разрядный счетчик T1.

Для работы таймера в этом режиме запрограммируем следующие действия: при нажатии на первую кнопку SW0 на вход таймера поступают сигналы с частотой $F_{cnt} = F_{ск} / 1024$, при нажатии на вторую кнопку SW1 – сигналы с частотой $F_{cnt} = F_{ск} / 256$. В обоих случаях сразу после нажатия загораются светодиоды, а после переполнения таймера и обработки соответствующего прерывания светодиоды гаснут. Таким образом, во втором случае время свечения светодиодов будет в 4 раза меньше, чем в первом, что и будет означать правильность работы программы.

Программа 4. 2

```
; *****
;Программа 4.2 для МК ATx8515: демонстрация работы таймера T1
;в режиме таймера. При частоте работы микроконтроллера Fск = 3,69 МГц
;после нажатия SW0 на вход счётчика поступают сигналы с частотой Fск/1024,
;после нажатия SW1 – Fск /256.
;Соединения: PD0,PD1-SW0,SW1; PB-LED
; *****

;.include "8515def.inc"           ;файл определений AT90S8515
;.include "m8515def.inc"         ;файл определений ATmega8515
.def temp = r16                  ;временный регистр
.equ SW0 = 0                     ;0-ой вывод порта PD
.equ SW1 = 1                     ;1-ий вывод порта PD
;***Таблица векторов прерываний
```

```
.org $000
    rjmp INIT                ;обработка сброса
.org $006
    rjmp T1_OVF              ;обработка переполнения таймера T0
;***Инициализация МК
INIT:    ldi temp,low(RAMEND) ;установка
        out SPL,temp        ; указателя стека
        ldi temp,high(RAMEND) ; на последнюю
        out SPH,temp        ; ячейку ОЗУ
        clr temp            ;инициализация выводов порта PD
        out DDRD,temp       ; на ввод
        ldi temp,0x03        ;включение 'подтягивающих' резисторов
        out PORTD,temp       ; в разрядах 0,1 порта PD
        ser temp            ;инициализация выводов порта PB
        out DDRB,temp       ; на вывод
        out PORTB,temp      ;выключение светодиодов
;***Настройка таймера T1 на режим таймера
        ldi temp,0x80        ;разрешение прерывания по
        out TIMSK,temp      ; переполнению таймера T1
        clr temp            ;таймер T1
        out TCCR1B,temp     ; остановлен
        ldi temp,0x80        ; загрузка TCNT1
        out TCNT1H,temp
        ldi temp,0x00
        out TCNT1L,temp
        sei                  ;глобальное разрешение прерываний

;***Ожидание нажатия кнопок
test_sw0: sbic PIND,SW0      ;проверка нажатия
        rjmp test_sw1       ; кнопки SW0
;***Обработка нажатия кнопки SW0
        ldi temp,0x05        ;для настройки предделителя (K=1024)
        rcall LED_ON        ;включение светодиодов
test_sw1: sbic PIND,SW1      ;проверка нажатия
        rjmp test_sw0       ; кнопки SW1
;***Обработка нажатия кнопки SW1
        ldi temp,0x04        ; для настройки предделителя (K=256)
```

```
rcall LED_ON          ;включение светодиодов
rjmp test_sw0

;***Включение светодиодов
LED_ON:  out TCCR1B,temp      ;запуск таймера с делителем
        clr temp             ;включение
        out PORTB,temp       ; светодиодов
        ret

;***Обработка прерывания при переполнении таймера T1
T1_OVF:  ser temp
        out PORTB,temp       ;выключение светодиодов
        clr temp             ;останов
        out TCCR1B,temp       ; таймера T1
        ldi temp,0x80
        out TCNT1H,temp       ; перезагрузка TCNT1
        ldi temp,0x00
        out TCNT1L,temp
        reti
```

На плате STK500 необходимо соединить выводы порта PD0, PD1 с кнопками SW0,SW1, шлейфом выводы порта PB – со светодиодами LED0 – LED7.

Задание 2. Проверить работу программы 4.2. Оценить время свечения светодиодов при нажатии кнопки SW0 и при нажатии кнопки SW1 и сравнить его с расчетным значением.

Изменив настройки таймера, уменьшить вдвое время включения светодиодов.

4. Программирование функций сравнения, захвата и ШИМ таймера T1

Функция сравнения

Подготовить программу для исследования функции сравнения таймера/счётчика T1. Возможный вариант работы программы: при нажатии на кнопку START (SW0) запускается таймер – происходит инкремент счётчика с частотой $F_{cnt}=F_{clk}/K$. При совпадении значений счётчика TCNT1 и регистра сравнения OCR1B происходит изменение состояния вывода OC1B на противоположное. При совпадении значений счётчика TCNT1 и регистра сравнения OCR1A происходит изменение состояния вывода OC1A на противоположное и сброс счётного регистра в нулевое состояние. Временные диаграммы работы таймера T1 приведены на рис.3.

Длительность выходных сигналов

$$t_u(OC1A) = t_u(OC1B) = OCR1A \cdot T_{cnt} = OCR1A \cdot K/F_{clk},$$

период формирования выходных сигналов

$$T = 2 t_u = 2 \cdot OCR1A \cdot K / F_{ск},$$

время задержки (сдвига)

$$t_3 = (OCR1A - OCR1B) \cdot K / F_{ск}.$$

Возможность останова таймера во время счёта реализуется с помощью внешнего прерывания от кнопки STOP (SW2). При записи значений в регистры сравнения необходимо соблюдать установленный порядок: сначала записывается старший байт, затем – младший.

Для наблюдения изменений состояний выводов OC1A и OC1B их необходимо соединить с выводами светодиодов: LED0 – PE2, LED1 – PD5. Подключения кнопок: SW0 – PD0, SW2 – PD2.




Рис.3. Временные диаграммы работы таймера/счётчика T1 при исследовании функции сравнения

Программа 4.3

```
;*****
;Программа 4.3 для МК ATx8515:
;демонстрация работы функции сравнения таймера T1.
;Частота тактового генератора Fск=3,69 МГц.
;При нажатии на SW0 (START) запускается счёт с частотой Fск/K,
;при нажатии на SW2 (STOP) счёт останавливается.
;При совпадении содержимого счётчика и регистра сравнения OCR1B
;переключается светодиод LED0,
;содержимого счётчика и регистра сравнения OCR1A – LED1.
;Соединения: LED0-PE2, LED1-PD5, SW0-PD0, SW2-PD2
;*****

#include "8515def.inc"           ;файл определений AT90S8515
#include "m8515def.inc"          ;файл определений ATmega8515
.def temp = r16                  ;временный регистр
.equ START = 0                   ;0-ой вывод порта PD
```

```
.org $000
    rjmp INIT          ;обработка сброса
.org $001
    rjmp STOP_PRESSED  ;обработка внешнего прерывания INT0 -
                        ; нажатие STOP
;***Инициализация МК
INIT:    ldi temp,low(RAMEND) ;установка
        out SPL,temp        ; указателя стека
        ldi temp,high(RAMEND) ; на последнюю
        out SPH,temp        ;   ячейку ОЗУ
        ldi temp,0x20        ;инициализация вывода PD5
        out DDRD,temp        ; как выхода
        ldi temp,0x05        ;включение 'подтягивающих' резисторов
        out PORTD,temp       ; в PD0, PD2
        ldi temp,0x04        ;/// для ATmega8515 инициализация вывода порта
        out DDRE,temp        ;/// PE2 (OC1B) на вывод
        ldi temp,(1<<INT0)   ;разрешение прерывания INT0
        out GICR,temp        ; в регистре GICR (или GIMSK)
        clr temp             ;обработка прерывания INT0
        out MCUCR,temp       ; по низкому уровню
;***Настройка функции сравнения таймера T1
        cli                  ;запрещение прерываний
        ldi temp,0x50        ;при сравнении состояния выводов OC1A и
        out TCCR1A,temp       ; OC1B изменяются на противоположные
        clr temp              ;останов
        out TCCR1B,temp       ; таймера
        ldi temp,0x08        ;запись числа в
        out OCR1BH,temp       ; регистр сравнения,
        ldi temp,0x00        ; первым записывается
        out OCR1BL,temp       ;   старший байт
        ldi temp,0x10        ;запись числа в
        out OCR1AH,temp       ; регистр сравнения,
        ldi temp,0x00        ; первым записывается
        out OCR1AL,temp       ;   старший байт
        clr temp              ;обнуление
        out TCNT1H,temp       ; счётного
        out TCNT1L,temp       ; регистра
```

```

        sei                                ;разрешение прерываний
WAITSTART: sbic PIND,START                ;ожидание нажатия
        rjmp WAITSTART                    ; кнопки START
        ldi temp,0x0D                     ;запуск таймера с предделителем K=1024,
        out TCCR1B,temp                   ; при совпадении с OCR1A – сброс
LOOP:    nop                              ;во время цикла происходит
        rjmp LOOP                        ; увеличение содержимого счётного регистра
;***Обработка прерывания от кнопки STOP
STOP_PRESSED:
        clr temp                          ;останов
        out TCCR1B,temp                   ; таймера
WAITSTART_2:                             ;ожидание
        sbic PIND,START                  ; нажатия
        rjmp WAITSTART_2                 ; кнопки START
        ldi temp,0x0D                     ;запуск
        out TCCR1B,temp                   ; таймера с предделителем K=1024
        reti

```

Задание 3. Проверить работу программы 4.3. Результат работы программы должен соответствовать диаграммам на рис.3. При нажатии на кнопку SW0 светодиоды работают в следующей последовательности: оба светодиода горят, далее выключается LED0, затем LED1, включается LED0, затем LED1 и т.д. В любой момент процесс можно остановить нажатием кнопки SW2.

Изменить параметры настройки таймера так, чтобы параметры выходных сигналов соответствовали выбранным значениям из ряда:

длительность $t_i \approx 2 \text{ с}, 3 \text{ с}, 4 \text{ с}, 5 \text{ с}, 6 \text{ с};$ задержка $t_z = \frac{1}{4} t_i, \frac{1}{2} t_i, \frac{3}{4} t_i.$

Режим ШИМ

Подготовить программу для исследования работы таймера T1 в режиме ШИМ. Выводы OC1A (PD5) и OC1B (PE2) необходимо подключить к светодиодам (PD5-LED0, PE2-LED1), выводы порта PD0, PD1 к кнопкам общего назначения SW0, SW1 соответственно. Варианты обработки нажатия кнопок:

- 1) при нажатии на SW0 формируются ШИМ-сигналы с порогом сравнения F1;
- 2) при нажатии на SW1 формируются ШИМ-сигналы с порогом сравнения F2;

Временная диаграмма работы ШИМ приведена на рис. 4 (T – период, t_i – длительность

выходных сигналов OC1A, OC1B).




Рис. 4. Формирование ШИМ-сигналов OC1A (неинвертированный) и OC1B (инвертированный)

Программа 4. 4

```
;*****
;Программа 4.4 для МК ATx8515:
;демонстрация работы таймера T1 в режиме ШИМ
;При нажатии SW0 происходит генерация
;ШИМ-сигналов с порогом сравнения F1
;При нажатии SW1 происходит генерация
;ШИМ-сигналов с порогом сравнения F2
;Соединения: PD5-LED0, PE2-LED1, PD0-SW0, PD1-SW1
;*****

.include "8515def.inc"          ;файл определений AT90S8515
.include "m8515def.inc"        ;файл определений ATmega8515
.def temp = r16                 ;временный регистр
;***Выходы порта PD
.equ SW0 = 0
.equ SW1 = 1

.org $000
    rjmp INIT                   ;обработка сброса
;***Инициализация МК
INIT:    ldi temp,0x20           ;инициализация PD5
        out DDRD,temp           ; на вывод
        ldi temp,0x03           ;включение 'подтягивающих'
        out PORTD,temp         ; резисторов порта PD
```

```
ldi temp,0x04    ;/// для ATmega8515 инициализация
out DDRE,temp    ;/// PE2 (OC1B) на вывод
cli              ;запрещение прерываний
;настройка таймера: 10-разрядный режим ШИМ, на выводе
;OC1A неинвертированный сигнал, OC1B - инвертированный сигнал
ldi temp,0xB3
out TCCR1A,temp
clr temp          ; обнуление
out TCNT1H,temp   ; счётного
out TCNT1L,temp   ; регистра
ldi temp,0x05     ;таймер
out TCCR1B,temp    ; запущен с делителем 1024

F1: sbic PIND,SW0    ;проверка нажатия SW0
    rjmp F2
;***Установка порога F1
ldi temp,0x00      ;запись числа в
out OCR1AH,temp     ; регистры сравнения,
out OCR1BH,temp     ; первым записывается
ldi temp,0xff       ; старший байт
out OCR1AL,temp
out OCR1BL,temp

F2: sbic PIND,SW1    ; проверка нажатия SW1
    rjmp F1
;*** Установка порога F2
ldi temp,0x02       ;запись числа в
out OCR1AH,temp     ; регистры сравнения,
out OCR1BH,temp     ; первым записывается
ldi temp,0xff       ; старший байт
out OCR1AL,temp
out OCR1BL,temp
rjmp F1
```

Задание 4. Проверить работу программы 4.4 на плате STK500. При нажатии на SW0 или SW1 светодиоды попеременно включаются/выключаются в соответствии с заданным порогом сравнения F1 или F2. Зарисовать диаграммы включения/выключения светодиодов LED0, LED1 с

учетом порогов сравнения, указав период и длительности сигналов.

Изменить программу для формирования двух выходных ШИМ-сигналов по заданному варианту работы:

- 1) с коэффициентом заполнения (отношение длительности к периоду) $\frac{1}{4}$, $\frac{1}{2}$, или $\frac{3}{4}$ для 9-разрядного режима ШИМ;
- 2) с коэффициентом заполнения $\frac{1}{4}$, $\frac{1}{2}$, или $\frac{3}{4}$ для 8-разрядного режима ШИМ;
- 3) с коэффициентом заполнения $\frac{1}{4}$ или $\frac{3}{4}$ для 10-разрядного режима ШИМ;
- 4) увеличив (или уменьшив) вдвое скважность (отношение периода к длительности) выходных сигналов по сравнению с запрограммированными.

Привести расчеты загружаемых значений OCR1A1, OCR1B, периода и длительности выходных сигналов, полагая частоту тактовых сигналов микроконтроллера $F_{ск} = 3,69$ МГц.

Функция захвата

Подготовить программу для исследования функции захвата таймера/счётчика T1. Захват должен происходить при нажатии соответствующей кнопки. По событию «захват» должен вызываться обработчик прерывания, который переписывает содержимое 16-разрядного регистра захвата ICR1 в регистры для хранения младшего и старшего байта. Пересылка данных в порт для индикации возможна только из регистров общего назначения, поэтому должны использоваться регистры хранения байтов. Вывод каждого байта можно связать с нажатием отдельной кнопки. С помощью регистра сравнения OCR1A можно задать максимальное значение, которое можно получить в счётчике и, следовательно, в регистре захвата. Для этого необходимо настроить работу таймера так, чтобы он сбрасывался при равенстве значений счётчика TCNT1 и регистра сравнения, установив бит CTC1 регистра TCCR1B в «1». Это будет дополнительным признаком правильности работы программы. Так например, если содержимое OCR1A равно \$7FFF, а бит CTC1 установлен в «1», то в старшем разряде регистра захвата всегда будет 0.

При моделировании программы в AVR Studio необходимо помнить, что вывод ICP в микроконтроллере AT90S8515 является выделенным, а в микроконтроллере ATmega8515 – это линия порта ввода/вывода PE0, которую необходимо запрограммировать на ввод. Вызов обработчика прерывания по событию «захват» происходит после установки в «1» флага TICIE1 регистра маски TIMSK и флага ICF1 регистра запросов TIFR.

Коммутация кнопок с выводами микроконтроллера: SW0 – PD0, SW1 – PD1, SW2 – PD2, SW3 – PE0. Светодиоды подключаются к выводам порта PB 10-проводным шлейфом.

Программа 4. 5

; *****

```
;Программа 4.5 для МК АТх8515:
;демонстрация работы функции захвата таймера T1.
;При нажатии SW0 (START) на вход счётчика поступает сигнал с частотой
; 3,69 МГц/1024, при нажатии SW3 (CAPT) происходит захват состояния таймера.
;При совпадении содержимого счётчика и регистра сравнения OCR1A
;происходит сброс таймера.
;При нажатии SW1 (SHOW_L) на светодиоды выводится значение младшего
;байта регистра захвата, SW2 (SHOW_H) – старшего байта регистра захвата.
;Соединения: SW0-PD0, SW1-PD1, SW2-PD2, SW3-PE0, 10-проводным шлейфом PB-LED
;*****
;.include "8515def.inc"           ;файл определений АТ90S8515
;.include "m8515def.inc"         ;файл определений АТмега8515
.def temp = r16                  ;временный регистр
.def H_byte = r17                ;для хранения старшего байта
.def L_byte = r18                ;для хранения младшего байта
.equ START = 0                   ;0-ой вывод порта PD
.equ SHOW_L = 1                  ;1-ый вывод порта PD
.equ SHOW_H = 2                  ;2-ой вывод порта PD
;***Векторы прерываний
.org $000
        rjmp INIT                ;обработка сброса
.org $003
        rjmp CAPT_PRESSED        ;обработка внешнего прерывания INT0 –
                                ; нажатие CAPT

;***Инициализация МК
INIT:    ldi temp, low(RAMEND)    ;установка
        out SPL, temp            ; указателя стека
        ldi temp, high(RAMEND)   ; на последнюю
        out SPH, temp            ; ячейку ОЗУ
        clr temp                 ;инициализация выводов
        out DDRD, temp           ; порта PD на ввод
        ldi temp, 0x07           ;включение 'подтягивающих'
        out PORTD, temp          ; резисторов порта PD
        clr temp                 ;/// для АТмега8515 инициализация
        out DDRE, temp           ;/// PE0 (ICP) на ввод
        ldi temp, 0x01           ;/// включение 'подтягивающего'
        out PORTE, temp          ;/// резистора порта PE0
```

```

ser temp                ;инициализация выводов
out DDRB,temp           ; порта PB на вывод
out PORTB,temp          ;выключение светодиодов
cli                     ;запрещение прерываний
clr temp                ;отключение от таймера
out TCCR1A,temp          ; выводов портов
clr temp                ;таймер
out TCCR1B,temp          ; остановлен
ldi temp,0xFF           ;запись числа в
out OCR1AH,temp          ; регистр сравнения,
ldi temp,0xFF           ; первым записывается
out OCR1AL,temp          ; старший байт
clr temp                ;обнуление
out TCNT1H,temp          ; содержимого
out TCNT1L,temp          ; счётного регистра
ldi L_byte,0x00          ; обнуление
ldi H_byte,0x00          ; регистров вывода
sei                     ;глобальное разрешение прерываний
WAITSTART: sbic PIND,START ;ожидание нажатия
rjmp WAITSTART          ; кнопки START
ldi temp,0x08           ;разрешение прерывания
out TIMSK,temp          ; по событию "захват" таймера
ldi temp,0xCD           ;запуск таймера с предделителем 1024,
out TCCR1B,temp          ; при совпадении с OCR1A - сброс
WAIT_L: sbic PIND,SHOW_L ;ожидание нажатия
rjmp WAIT_H             ; кнопки SHOW_L - показать младший байт
out PORTB,L_byte         ;вывод на СД
WAIT_H: sbic PIND,SHOW_H ;ожидание нажатия
rjmp WAIT_L             ; кнопки SHOW_H - показать старший байт
out PORTB,H_byte         ;вывод на СД
rjmp WAIT_L

;***Обработка прерывания от кнопки CAPT
CAPT_PRESSED:
in L_byte,ICR1L          ;считывание младшего байта
in H_byte,ICR1H          ;считывание старшего байта
com L_byte               ;инвертирование для вывода
com H_byte               ;инвертирование для вывода

```


reti

Задание 5. Загрузить программу в STK500. Для проверки работы программы включите одновременно секундомер часов и запустите программу. После останова программы проверьте показания времени на часах и в регистрах захвата таймера. После сброса RESET повторите эксперимент (2-3 раза) и оцените погрешность замеров.

Оформление отчета

Отчет должен содержать:

- а) схемы таймеров T0 и T1;
- б) тексты переработанных программ по заданиям 1 и 2;
- в) программы для проверки функций сравнения и формирования ШИМ-сигналов по выбранным вариантам, временные диаграммы выходных сигналов. На диаграммах указать временные характеристики сигналов (T , t_n , t_z). Привести расчеты пороговых значений F1 и F2 для формируемых ШИМ-сигналов;
- г) программу функции захвата с оценками наблюдений;
- д) цели работы и выводы.

Требования при защите: уметь отвечать на вопросы по схемам таймеров, программированию отдельных функций.

Контрольные вопросы

1. В каком порядке выполняется запись и чтение байтов в 16 - разрядные регистры таймера T1?
2. Как подключены светодиоды к выводам порта микроконтроллера? Сигнал какого уровня (логического 0 или логической 1) включает светодиод?
3. Какова длительность сигналов, формируемых программой 4.3? Какими параметрами настройки таймера определяются период и величина относительного сдвига по времени выходных сигналов OC1B и OC1A?
4. Какова длительность сигналов, формируемых программой 4.4? Какие значения необходимо занести в регистры сравнения для формирования ШИМ - сигналов скважностью 4 при том же периоде ШИМ - сигналов?
5. Как изменится скважность ШИМ - сигналов, формируемых при работе программы 4.4, если изменить управляющее слово в регистре TCCR1A на 0xB1 (0xB2)?
6. Назовите максимальное время захвата, которое можно зарегистрировать при работе

программы 4.5; при рабочей частоте микроконтроллера 1 МГц. В каких приложениях можно применить эту программу?

7. В каких разрядах регистра захвата регистрируются секундные интервалы времени при частоте работы микроконтроллера 3,69 МГц с коэффициентом делителя 1024?

8. Подготовить программу для выполнения **Задания 1** при условии сохранения режима процессора sleep, проверяя замыкание кнопки и выполняя ввод сигнала в таймер с помощью обработчика внешнего прерывания.

9. Разработать модели для виртуального моделирования таймерных режимов сравнения и ШИМ в среде PROTEUS. Подключить цифровой осциллограф и пробники для контроля сигналов на выходах OC1A, OC1B. Зафиксировав кнопку режима работы (SW0 или SW1) в замкнутом состоянии, выполнить симуляцию в окне Digital Analyses и проверить период и длительность формируемых сигналов при частоте работы микроконтроллера 1 МГц. Подключив светодиоды к выходам схемы проверить их работу на модели.